

Лабораторная работа № 4 (дополнительная)

Решение статической плоской задачи электроупругости «Статическое деформирование пьезокерамического преобразователя с многоэлектродным покрытием под действием заданной разности потенциалов на электродах»

Цель работы:

1. Изучить основы программирования на языке APDL программного комплекса ANSYS на примере решения статической осесимметричной задачи для пьезоэлектрического диска с многоэлектродным покрытием (файл FEM_PD_St1.inp):
 - а) задание констант пьезоэлектрического материала;
 - б) элементные системы координат и их применение для задания анизотропных и пьезоэлектрических свойств;
 - в) использование регулярного разбиения
 - г) задание электродированных поверхностей и граничных условий на них.
2. Составить программу на языке APDL ANSYS для индивидуальной статической задачи для пьезоэлектрического диска с многоэлектродным покрытием.
3. Проанализировать результаты и оформить отчет.

Пример

Пьезоэлектрический диск радиуса R и толщины H отнесен к цилиндрической системе координат Orz ($0 \leq r \leq R, -H/2 \leq z \leq H/2$) и находится в условиях осесимметричного деформирования. В силу осесимметричности задачи далее рассматривается только его меридиональное сечение, причем в соответствие с методологией ANSYS ось r считается осью X , а ось z – осью Y .

Диск имеет четыре электродированные поверхности (рис. 1): электрод с номером 1 – $0 \leq X \leq R_1, Y = -H/2$; электрод с номером 2 – $0 \leq X \leq R_1, Y = H/2$; электрод с номером 3 – $R_2 \leq X \leq R, Y = -H/2$; электрод с номером 4 – $R_2 \leq X \leq R, Y = H/2$. Пьезоэлектрический диск выполнен из пьезокерамики PZT-4, причем зоны $\{0 \leq X \leq R_1, -H/2 \leq Y \leq H/2\}$ и $\{R_1 \leq X \leq R_2, -H/2 \leq Y \leq H/2\}$ поляризованы вдоль оси Y , а зона $\{R_2 \leq X \leq R, -H/2 \leq Y \leq H/2\}$ поляризована противоположно оси Y (Рис. 2).




Рис. 1 Схема пьезоэлектрического диска




Рис. 2 Схема меридионального сечения пьезоэлектрического диска

Диск закреплен по оси Y в зоне $X = R, Y = 0$, т.е. $U_y = 0$ при $X = R, Y = 0$, на оси $X = 0$ следует задать условия симметрии, а остальные поверхности диска свободны от механических напряжений.

Диск деформируется под действием подаваемого электрического напряжения на электроды: $\varphi = -V_{inp}$ – на первом электроде, $\varphi = V_{inp}$ – на втором электроде, $\varphi = -V_{inp}$ – на третьем электроде, и $\varphi = V_{inp}$ – на четвертом электроде.

Необходимо определить смещения диска в результате деформирования.

РЕШЕНИЕ ЗАДАЧИ С ИСПОЛЬЗОВАНИЕМ ANSYS

Пример решения задачи с использованием программы ANSYS содержится в файле FEM_PD_St1.inp, листинг которого приведен ниже. Программа предназначена для выполнения в пакетном режиме ANSYS и снабжена подробными комментариями. Другой пример решения подобной задачи содержится также в файле Vm142.dat из набора фирменных примеров, предоставляемых ANSYS

Начинать работу с программным комплексом ANSYS лучше с запуска ANSYS Product Launcher (Пуск→Все программы), где следует задать среду моделирования (Simulation Environment→ANSYS), выбрать рабочий каталог, в который будут сохраняться все рабочие файлы ANSYS (Working Directory) и задать название проекта (Job Name).

Выполнение *командного файла* (текстовый файл с расширением .inp или .txt), написанного на языке APDL ANSYS осуществляется с помощью пункта меню File → Read Input from... В процессе создания командного файла можно копировать блоки команд в командную строку и смотреть на результат выполнения.

Периодически следует сохранять результаты работы! Перед запуском новых расчетов надо очищать текущую базу данных (File→Clear and Start New).

Листинг входного файла FEM_PD_St1.inp

```
! (все, что следует после восклицательного знака, игнорируется
интерпретатором, это комментарии!)
! Файл FEM_PD_St1.inp
! Тестовая задача № 1
! О С Е С И М М Е Т Р И Ч Н А Я   З А Д А Ч А
! П Ъ Е З О Э Л Е К Т Р И Ч Е С К И Й   Д И С К
! Статическая задача

! задаем заголовок для решаемой задачи ("пьезоэлектрическая пластина")
```

/TITLE, PIEZOELECTRIC PLATE

! Задание параметров для геометрических размеров (все - в системе СИ)

H=0.002 ! толщина диска

R=10*H ! радиус диска

! Задание параметров для вспомогательных геометрических размеров

R1=R/4 ! радиус внутренней электродированной поверхности

R23=R/4 ! размер по радиусу для кольцевой электродированной поверхности

! Задание параметра для значения подаваемого электрического потенциала на электрод

VINP=1

! Задание параметров триангуляции (размеров конечных элементов)

HDIV=8 ! Кол-во КЭ по толщине диска (должно быть четным числом!)

R1DIV=16 ! Кол-во КЭ по радиальному направлению от 0 до R1

R12DIV=32 ! Кол-во КЭ по радиальному направлению от R1 до R2

R23DIV=16 ! Кол-во КЭ по радиальному направлению от R2 до R

! Задание параметров для материальных констант пьезокерамики PZT-4

! (все данные - в системе СИ)

RO=7.5e3 ! плотность

C11E=13.9e10 ! упругие модули C^E_{ij}

C12E=7.78e10

C13E=7.43e10

C33E=11.5e10

C66E=(C11E-C12E)/2.

C44E=2.56e10

E31=-5.2 ! пьезомодули

E33=15.1

E15=12.7

EPS0=8.85e-12 ! диэлектрическая проницаемость вакуума

EPS11=730*EPS0 ! диэлектрические проницаемости

EPS33=635*EPS0

/PREP7 ! (Preprocessor) Вход в препроцессор

! Занесение материальных констант пьезокерамического материала в массивы данных

! (для случая осесимметричной задачи)

MP,DENS,1,RO

TB,ANEL,1

TBDATA,1,C11E,C13E,C12E

TBDATA,7,C33E,C13E

TBDATA,12,C11E

TBDATA,16,C44E

TB,PIEZ,1

TBDATA,2,E31

TBDATA,5,E33

TBDATA,8,E31

TBDATA,10,E15

MP,PERX,1,EPS11

MP,PERY,1,EPS33

! Задание типов конечных элементов, используемых для решения задачи

ET,1,PLANE13,7,,1 ! (Element Type) 1й аргумент - номер типа конечного эл

! 2й аргумент - название используемого конечного элемента (PLANE13 - четырехузловой плоский КЭ)

! 3й аргумент - значение опции KEYOPT(1) из описания конечного элемента (здесь равно 7)

! для элемента PLANE13 KEYOPT(1)=7 задает три степени свободы UX,UY,VOLT

! UX - перемещение по оси OX, x-компонента вектора перемещений

! UY - перемещение по оси OY, y-компонента вектора перемещений

! VOLT - электрический потенциал

! 4й аргумент - значение опции KEYOPT(2) из описания конечного элемента (здесь используем значение по умолчанию)
! 5й аргумент - значение опции KEYOPT(3) из описания конечного элемента (здесь равно 1)
! для элемента PLANE13 KEYOPT(3)=1 задает осесимметричный КЭ

R2=R-R23 ! Вспомогательный геометрический размер по радиусу

! Определение опорных точек

K,1,0,-H/2 ! (Keypoint) точка с номер 1 с координатами x=0, y=-H/2
K,2,R1,-H/2
K,3,R2,-H/2
K,4,R,-H/2
K,5,R,H/2
K,6,R2,H/2
K,7,R1,H/2
K,8,0,H/2

! Определение прямых линий, соединяющих точки

L,1,2 ! (Line) линия, соединяющая точки 1 и 2 (будет иметь номер 1)
L,2,3 \$ L,3,4 ! линии 2 и 3, знак \$ - для разделения команд, записанных в одной строке
L,4,5 \$ L,5,6 \$ L,6,7 \$ L,7,8 \$ L,8,1 \$ L,6,3 ! линии 4-9

! Построение областей (площадей) по опорным точкам

! точки должны быть перечислены в порядке обхода, по или против часовой стрелки
A,1,2,3,6,7,8 ! (Area) область 1
A,3,4,5,6 ! область 2

! Определение локальной системы координат с номером 11,

! повернутой относительно оси z на 180 градусов

LOCAL,11,0,,,,180 ! 1й аргумент - пользовательский номер коорд. системы (наименьший доступный номер равен 11),
! 2й аргумент - номер глобальной коорд. системы, на основе кот. строится пользовательская (0 - декартова)
! 3й аргумент - x-координата начала новой СК в глобальной декартовой СК (по умолчанию 0, значение по умолчанию может быть опущено)
! 4й аргумент - y-координата начала новой СК в глобальной декартовой СК (по умолчанию 0)
! 5й аргумент - z-координата начала новой СК в глобальной декартовой СК (по умолчанию 0)
! 6й аргумент - угол вращения относительно локальной оси z от положительного направления оси x к y (здесь 180)
! 7й аргумент - угол вращения относительно локальной оси x от положительного направления оси y к z (здесь отсутствует, так как равен 0)
! 8й аргумент - угол вращения относительно локальной оси y от положительного направления оси z к x (здесь отсутствует, так как равен 0)

CSYS,0 ! (change system) переход в систему координат под номером 0 (глобальную декартову)

! Зададим для области 2 элементную систему координат 11 (для задания направления поляризации)

! Сначала выберем область 2

ASEL,S,AREA,,2 ! (Area Select) 1й аргумент - тип набора для выбора (здесь S - новый набор)
! 2й аргумент - метка набора, выбирается из имеющихся вариантов (по умолчанию для команды ASEL это название сущности AREA)
! 3й аргумент - компонент метки (не используется для AREA)
! 4й аргумент - минимальное значение диапазона данных
! для AREA задается номер области (здесь 2)
! 5й аргумент - максимальное значение диапазона данных (по умолчанию совпадает с минимальным значением)

! 6й аргумент - шаг для занесения данных диапазон (по умолчанию равен 1)

! задаем атрибут (номер элементной системы координат) для выбранной области
! AATT - задание специальных атрибутов для еще не разбитых на конечные элементы областей
AATT,1,,1,11 ! 1й аргумент - номер материала для выбранной области (в задаче используется один материала, по умолчанию его номер равен 1)
! 2й аргумент - номер набора материальных констант конечного элемента (здесь не используется)
! 3й аргумент - номер типа КЭ для выбранной области (здесь 1)
! 4й аргумент - номер элементной системы координат для выбранной области (здесь 11)

!необязательные команды
ASEL,S,AREA,,1
AATT,1,,1

ASEL,ALL ! возврат к выбору всех областей

! Разбиение линий для канонической конечно-элементной сетки
! Важно: суммарное число делений для противоположных линий области должно быть одинаковым!

! Сначала выберем все вертикальные линии (это линии L8, L9, L4)
LSEL,S,LOC,X,0 ! (Line Select) 1й аргумент - тип набора для выбора (здесь S - новый набор)
! 2й аргумент - метка набора, выбирается из имеющихся вариантов (здесь LOC)
! 3й аргумент - компонент метки (для метки LOC это X, Y или Z)
! 4й аргумент - минимальное значение диапазона данных
! для LOC=X задается значение координаты X (здесь X=0)
! линия, лежащая на прямой X=0 - это линия 8
! 5й аргумент - максимальное значение диапазона данных (по умолчанию совпадает с минимальным значением)
! 6й аргумент - шаг для занесения данных диапазон (по умолчанию равен 1)

LSEL,A,LOC,X,R2 ! Добавим в набор линию, лежащую на прямой X=R2; это линия L9
! 1й аргумент - тип набора для выбора (здесь A - добавить в существующий набор)
LSEL,A,LOC,X,R ! Добавим в набор линию, лежащую на прямой X=R; это линия L4

!-----
! Способ выбора тех же трех линий L8, L9, L4 по их номерам
LSEL,S,LINE,,8,9 (Line Select) 1й аргумент - тип набора для выбора (здесь S - новый набор)
! 2й аргумент - метка набора, выбирается из имеющихся вариантов (по умолчанию для команды LSEL это название сущности LINE)
! 3й аргумент - компонент метки (не используется для LINE)
! 4й аргумент - минимальное значение диапазона данных
! для LINE задается номер линии (здесь 8)
! 5й аргумент - максимальное значение диапазона данных (здесь 9)
! 6й аргумент - шаг для занесения данных диапазон (по умолчанию равен 1)

LSEL,A,LINE,,4 (Line Select) 1й аргумент - тип набора для выбора (здесь A - добавление к имеющемуся набору)
!-----

! Для всех выбранных трех линий зададим разбиение на одно и то же число частей
! LESIZE - задание числа делений и коэффициента разбиения для неразбитых линий
LESIZE,ALL,,,HDIV ! (Line Element Size) 1й аргумент - номер линии (здесь ALL - все выбранные ранее линии)
! 2й аргумент - длина стороны конечного элемента на линии, (задается, если отсутствует 4й аргумент - число делений)

! 3й аргумент - дуга разбиения в градусах (только для кривых линий)
! 4й аргумент - число делений на линии (здесь равно HDIV)
! 5й аргумент - коэффициент сгущения сетки, позволяет получить неравномерное разбиение (по умолчанию равен 1)

LSEL,S,LOC,X,0,R1 ! Выбор линий с координатой $0 \leq X \leq R1$, это линии L1 и L7
LESIZE,ALL,,,R1DIV ! для выбранных линий задается число делений R1DIV

LSEL,S,LOC,X,R1,R2 ! Выбор линий с координатой $R1 \leq X \leq R2$, это линии L2 и L6
LESIZE,ALL,,,R12DIV ! для выбранных линий задается число делений R2DIV

LSEL,S,LOC,X,R2,R ! Выбор линий с координатой $R2 \leq X \leq R$, это линии L3 и L5
LESIZE,ALL,,,R23DIV ! для выбранных линий задается число делений R23DIV

LSEL,ALL ! возврат к выбору всех линий

! Превращение нерегулярной области 1 в регулярную
LCCAT,1,2 ! Конкатенация линий L1 и L2, в результате будет линия L10
LCCAT,7,6 ! Конкатенация линий L7 и L6, в результате будет линия L11

MSHKEY,1 ! (Mesh Key) задает ключ разбиения (0 - свободное разбиение, 1 - регулярное разбиение)

MSHAPE,0,2 ! (Mesh Shape) задает форму конечных элементов

! 1й аргумент - ключ формы элементов (здесь 0 - разбиение на четырехугольные элементы)

! 2й аргумент - размерность модели (здесь 2D - разбиение областей)

AMESH,ALL ! (Area Mesh) Команда построения конечно-элементной сетки на областях

FINISH ! выход из процессора

/SOLU ! (Solution) Вход в решатель

ANTYPE,STAT ! (Analysis Type) Выбор типа анализа (здесь STATic - решение статической задачи)

!Определение электродов

! Выбор узлов для первого электрода (с координатами $0 \leq X \leq R1$, $Y = -H/2$)

NSEL,S,LOC,Y,-H/2 ! (Node Select)

! 1й аргумент - тип набора для выбора (здесь S - новый набор)

! 2й аргумент - метка набора, выбирается из имеющихся вариантов (здесь LOC)

! 3й аргумент - компонент метки (для метки LOC это X, Y или Z, здесь Y)

! 4й аргумент - минимальное значение диапазона данных

! для LOC=Y задается значение координаты Y (здесь $Y = -H/2$)

NSEL,R,LOC,X,0,R1 ! (Node Select) 1й аргумент - тип набора для выбора (здесь R - дополнительный выбор из имеющегося набора)

! 2й аргумент - метка набора, выбирается из имеющихся вариантов (здесь LOC)

! 3й аргумент - компонент метки (для метки LOC это X, Y или Z, здесь X)

! 4й аргумент - минимальное значение диапазона данных

! для LOC=X задается значение координаты X (здесь $X = R1$)

!Объединение всех выбранных узлов в одну группу с номером 1

CP,1,VOLT,ALL ! (Coupled node set) Определение связанного набора степеней свободы

! 1й аргумент - номер набора (по умолчанию равен 1)

! 2й аргумент - название степени свободы (здесь VOLT - электрический потенциал)

! 4й по 17й аргумент - список номеров узлов (здесь ALL - все узлы)

! задание параметра N_VOLT1, значение которого равно минимальному номеру узла для узлов группы 1

! узел с номером N_VOLT1 будет ссылочным узлом группы 1

***GET,N_VOLT1,NODE,,NUM,MIN** ! задает параметр со значением из расчетной базы данных

```

! 1й аргумент - пользовательское имя параметра (здесь N_VOLT1)
! 2й аргумент - имя сущности (здесь NODE)
! 3й аргумент - номер сущности (по умолчанию - выбор всех номеров заданной
сущности)
! 4й аргумент - компонент сущности, выбирается из доступных (здесь NUM -
номер сущности)
! 5й аргумент - метка компонента (для NUM это MIN или MAX)

! Выбор узлов для второго электрода (с координатами  $0 \leq X \leq R1$ ,  $Y = H/2$ )
NSEL,S,LOC,Y,H/2
NSEL,R,LOC,X,0,R1

!Объединение всех выбранных узлов в одну группу с номером 2
CP,2,VOLT,ALL

! задание параметра N_VOLT2, значение которого равно минимальному номеру узла
для узлов группы 2
! узел с номером N_VOLT2 будет ссылочным узлом группы 2
*GET,N_VOLT2,NODE,,NUM,MIN

! Выбор узлов для третьего электрода (с координатами  $R2 \leq X \leq R$ ,  $Y = -H/2$ )
NSEL,S,LOC,Y,-H/2
NSEL,R,LOC,X,R2,R

! Объединение всех выбранных узлов в одну группу с номером 3
CP,3,VOLT,ALL

! задание параметра N_VOLT3, значение которого равно минимальному номеру узла
для узлов группы 3
! узел с номером N_VOLT3 будет ссылочным узлом группы 3
*GET,N_VOLT3,NODE,,NUM,MIN

! Выбор узлов для четвертого электрода (с координатами  $R2 \leq X \leq R$ ,  $Y = H/2$ )
NSEL,S,LOC,Y,H/2
NSEL,R,LOC,X,R2,R

! Объединение всех выбранных узлов в одну группу с номером 4
CP,4,VOLT,ALL

! задание параметра N_VOLT4, значение которого равно минимальному номеру узла
для узлов группы 4
! узел с номером N_VOLT4 будет ссылочным узлом группы 4
*GET,N_VOLT4,NODE,,NUM,MIN

NSEL,ALL ! Возврат к выбору всех узлов модели.

! В задаче три степени свободы: UX,UY,VOLT
! Зададим электрические граничные условия

! Зададим значения потенциалов на электродах
D,N_VOLT1,VOLT,-VINP !(Degree of freedom (DOF) constraints) задание
ограничений на степени свободы в узлах
! 1й аргумент - номер узла (здесь ссылочный номер первой группы N_VOLT1)
! 2й аргумент - метка степени свободы (здесь VOLT - электрический потенциал)
! 3й аргумент - значение для заданной метки (здесь -VINP)
D,N_VOLT2,VOLT,VINP
D,N_VOLT3,VOLT,-VINP
D,N_VOLT4,VOLT,VINP

! Зададим механические граничные условия

! Зададим условия симметрии относительно оси OY (на этой оси лежит линия 8)
DL,8,,SYMM ! (DOF Line constraints) задание ограничений на степени свободы на
линиях

```

! 1й аргумент - номер линии
 ! 2й аргумент - номер области, содержащей линию (здесь используется значение по умолчанию)
 ! 3й аргумент - метка для условий симметрии или метка степени свободы (здесь SYMM - генерировать условия симметрии)

! Зададим условие закрепления срединной крайней точки по оси Y
 ! выберем узел с координатами X=R, Y=0

NSEL,S,LOC,Y,0

NSEL,R,LOC,X,R

D,ALL,UY,0 !(Degree of freedom (DOF) constraints) задание ограничений на степени свободы в узлах

! 1й аргумент - номер узла (здесь ALL - все выбранные узлы)
 ! 2й аргумент - метка степени свободы (здесь UY - перемещение по оси OY)
 ! 3й аргумент - значение для заданной метки (здесь UY=0)

NSEL,ALL ! Возврат к выбору всех узлов

SOLVE ! Решить систему МКЭ

FINISH ! Выход из процессора

! Вход в общий постпроцессор

/POST1

! Сервисная команда для расположения шкалы значений справа

/PLOPTS,INFO,AUTO

! Вывод картины распределения перемещений UY

PLNSOL,U,Y

Дадим дополнительные комментарии к программе.

Геометрические размеры, параметры триангуляции (определяющие размеры конечных элементов), материальные константы задаются в виде пользовательских параметров. Далее в препроцессоре (вход в препроцессор осуществляется командой **\PREP7**) с помощью команд **MP**, **TB**, **TBDATA** задаются матрицы упругих модулей, пьезомодулей и диэлектрических проницаемостей.

Определение материальных констант пьезоэлектрических материалов. Как известно, пьезоэффект наблюдается лишь в кристаллах, не имеющих центральной симметрии. В связи с этим, пьезоэлектрические тела обязательно должны обладать анизотропными свойствами. В общем случае в ANSYS для задания материальных констант пьезоэлектрических тел необходимо

определить следующие величины: плотность ρ ; симметричную матрицу упругих модулей $c_{\alpha\beta}^E$;

$\alpha, \beta = 1, \dots, 6$; $c_{\alpha\beta}^E = c_{\beta\alpha}^E$; матрицу пьезомодулей $e_{i\alpha}$; $i = 1, 2, 3$; $\alpha = 1, \dots, 6$; и диагональную

матрицу диэлектрических проницаемостей ε_{ii}^S ; $i = 1, 2, 3$. (Можно задавать и альтернативные

наборы констант типа матрицы упругих податливостей $s_{\alpha\beta}^E$ и др.)

Опишем технику определения в ANSYS модулей $c_{\alpha\beta}^E$ и $e_{i\alpha}$ более подробно.

Коэффициенты $c_{\alpha\beta}^E$ задаются в форме 6x6 матрицы (4x4 – для двумерных задач). В силу симметрии матрицы упругих модулей используется только ее верхняя треугольная часть, причем при нестандартном расположении коэффициентов:

$$\mathbf{c}_{\text{ANSYS,3D}} = \begin{array}{c|cccccc} & x & y & z & xy & yz & xz \\ \hline x & c_{11}^E & c_{12}^E & c_{13}^E & c_{16}^E & c_{14}^E & c_{15}^E \\ y & & c_{22}^E & c_{23}^E & c_{26}^E & c_{24}^E & c_{25}^E \\ z & & & c_{33}^E & c_{36}^E & c_{34}^E & c_{35}^E \\ xy & & & & c_{66}^E & c_{46}^E & c_{56}^E \\ yz & & \text{sym} & & & c_{44}^E & c_{45}^E \\ xz & & & & & & c_{55}^E \end{array} \quad (1.1)$$

$$\mathbf{c}_{\text{ANSYS,2D}} = \begin{array}{c|cccc} & x & y & z & xy \\ \hline x & c_{11}^E & c_{12}^E & c_{13}^E & c_{16}^E \\ y & & c_{22}^E & c_{23}^E & c_{26}^E \\ z & & & c_{33}^E & c_{36}^E \\ xy & \text{sym} & & & c_{66}^E \end{array} \quad (1.2)$$

где $c_{\alpha\beta}^E$ – общепринятые в теории пьезоэлектричества обозначения для упругих модулей.

Матрицы $\mathbf{c}_{\text{ANSYS,3(2)D}}$ в ANSYS задаются обычно командами: `TB,ANISO,MAT` и `TBDATA,STLOC,C1,C2,...,C6`; где `MAT` – номер набора материальных свойств, `STLOC` – номер для начала последовательного размещения данных `C1,C2,...,C6` в одномерном массиве данных. Массив данных для коэффициентов, входящих в (1.1), (1.2), заполняется по строкам (1.1) в виде одномерного массива с 21 компонентой:

$$\begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 6 \\ & 7 & 8 & 9 & 10 & 11 \\ & & 12 & 13 & 14 & 15 \\ & & & 16 & 17 & 18 \\ & & & & 19 & 20 \\ & & & & & 21 \end{bmatrix} \quad (1.3)$$

В результате получается следующее соответствие между данными из массива, определенного

`TBDATA`, и модулями упругости $c_{\alpha\beta}^E$:

№	1	2	3	4	5	6	7	8	9	10	
$c_{\alpha\beta}^E$	c_{11}^E	c_{12}^E	c_{13}^E	c_{16}^E	c_{14}^E	c_{15}^E	c_{22}^E	c_{23}^E	c_{26}^E	c_{24}^E	
№	11	12	13	14	15	16	17	18	19	20	21
	c_{25}^E	c_{33}^E	c_{36}^E	c_{34}^E	c_{35}^E	c_{66}^E	c_{46}^E	c_{56}^E	c_{44}^E	c_{45}^E	c_{55}^E

Для наиболее распространенного типа пьезоэлектрических материалов – пьезокерамики, поляризованной в направлении оси Oz , различные ненулевые модули упругости в массиве `TBDATA` будут иметь следующие номера (для пьезокерамики $c_{22}^E = c_{11}^E$; $c_{23}^E = c_{13}^E$; $c_{55}^E = c_{44}^E$; $c_{66}^E = (c_{11}^E - c_{12}^E)/2$):

№	1	2	3	7	8	12	16	19	21
$c_{\alpha\beta}^E$, пьезокерамик а	c_{11}^E	c_{12}^E	c_{13}^E	c_{11}^E	c_{13}^E	c_{33}^E	c_{66}^E	c_{44}^E	c_{44}^E

Для двумерных задач, как видно из сравнения (1.1) – (1.3), достаточно задавать только первые 16 позиций в одномерном массиве модулей $c_{\alpha\beta}^E$.

Заметим, что для плоских и осесимметричных двумерных задач обычно удобно считать в рабочей плоскости Oxy ось Oz направлением предварительной поляризации пьезокерамики (т.е. осью $z=3$) для модулей $c_{\alpha\beta}^E$). В этом случае, для плоских и осесимметричных задач различные ненулевые модули упругости в массиве TBDATA будут иметь номера:

№	1	2	3	7	8	12	16
$c_{\alpha\beta}^E$, пьезокерамика	c_{11}^E	c_{13}^E	c_{12}^E	c_{33}^E	c_{13}^E	c_{11}^E	c_{44}^E

Аналогичным образом в нестандартном формате задаются в ANSYS и пьезомодули $e_{i\alpha}$. Пьезомодули располагаются в следующем порядке в матрицах размера 6×3 (4×2 – для двумерных задач):

$$\mathbf{e}_{\text{ANSYS,3D}} = \begin{array}{c|ccc} & x & y & z \\ \hline x & e_{11} & e_{21} & e_{31} \\ y & e_{12} & e_{22} & e_{32} \\ z & e_{13} & e_{23} & e_{33} \\ xy & e_{16} & e_{26} & e_{36} \\ yz & e_{14} & e_{24} & e_{34} \\ xz & e_{15} & e_{25} & e_{35} \end{array} ; \quad \mathbf{e}_{\text{ANSYS,2D}} = \begin{array}{c|cc} & x & y \\ \hline x & e_{11} & e_{21} \\ y & e_{12} & e_{22} \\ z & e_{13} & e_{23} \\ xy & e_{16} & e_{26} \end{array}$$

После команды TB,PIEZ,MAT массив данных TBDATA для пьезомодулей в ANSYS заполняется из массива $\mathbf{e}_{\text{ANSYS,3D}}$ по строкам, как одномерный массив размера 18. В результате получается следующее соответствие между данными из массива TBDATA и пьезомодулями $e_{i\alpha}$:

№	1	2	3	4	5	6	7	8	9
$e_{i\alpha}$	e_{11}	e_{21}	e_{31}	e_{12}	e_{22}	e_{32}	e_{13}	e_{23}	e_{33}
№	10	11	12	13	14	15	16	17	18
$e_{i\alpha}$	e_{16}	e_{26}	e_{36}	e_{14}	e_{24}	e_{34}	e_{15}	e_{25}	e_{35}

причем для двумерных задач достаточно задавать элементы с номерами 1,2,4,5,7,8,10 и 11.

Для пьезокерамики, поляризованной в направлении оси Oz , различные ненулевые пьезомодули для трехмерных задач будут иметь следующие номера в массиве TBDATA (для пьезокерамики $e_{32} = e_{31}$; $e_{24} = e_{15}$):

№	3	6	9	14	16
$e_{i\alpha}$,	e_{31}	e_{31}	e_{33}	e_{15}	e_{15}

пьезокерамика					
---------------	--	--	--	--	--

Наконец, для плоских и осесимметричных двумерных задач в случаях, когда в плоскости Oxy ось Oy является осью предварительной поляризации пьезокерамики (т.е. осью $z=(3)$ для пьезомодулей $e_{i\alpha}$), различные ненулевые пьезомодули в массиве TBDATA будут иметь номера, указанные в следующей таблице:

№	2	5	8	10
$e_{i\alpha}$, пьезокерамика	e_{31}	e_{33}	e_{31}	e_{15}

Суммируя изложенное выше, для пьезокерамического материала можно написать следующие фрагменты программ для ANSYS, задающие наборы материальных констант MAT с номером 1 для трехмерных (3D) и двумерных (2D) задач, причем в последнем случае пьезокерамика считается поляризованной в рабочей плоскости Oxy вдоль оси Oy :

Фрагмент 1 (3D).

```
MP, DENS, 1, RHO
TB, ANEL, 1
TBDATA, 1, C11E, C12E, C13E
TBDATA, 7, C11E, C13E
TBDATA, 12, C33E
TBDATA, 16, C66E
TBDATA, 19, C44E
TBDATA, 21, C44E
TB, PIEZ, 1
TBDATA, 3, E31
TBDATA, 6, E31
TBDATA, 9, E33
TBDATA, 14, E15
TBDATA, 16, E15
MP, PERX, 1, EPS11
MP, PERZ, 1, EPS33
```

Фрагмент 2 (2D).

```
MP, DENS, 1, RHO
TB, ANEL, 1
TBDATA, 1, C11E, C13E, C12E
TBDATA, 7, C33E, C13E
TBDATA, 12, C11E
TBDATA, 16, C44E
TB, PIEZ, 1
TBDATA, 2, E31
TBDATA, 5, E33
TBDATA, 8, E31
TBDATA, 10, E15
MP, PERX, 1, EPS11
MP, PERY, 1, EPS33
```

Здесь $RHO=\rho$, $C11E=C_{11}^E$ и т.п. Эти величины являются скалярными параметрами и должны быть определены раньше. Кроме того, в представленных фрагментах добавлены команды MP, задающие плотность и диэлектрические проницаемости пьезокерамики.

Как видно, задание констант пьезоэлектрических материалов требует достаточных усилий. Важно подчеркнуть, что в представленных фрагментах для определяемых наборов констант оси $Oxy(z)$ являются осями элементных систем координат!

Команда **ET,1,PLANE13,7,,1** задает четырехузловой конечный элемент PLANE13 с опцией пьезоэлектрического анализа и опцией осесимметричности, который будет использован для решения задачи. Для пользователя все отличие осесимметричной задачи от двумерной задачи о плоской деформации состоит только в третьей опции данного конечного элемента. (Если записать **ET,1,PLANE13,7,,0** то будет решаться двумерная задача о плоской деформации.) Однако ANSYS будет при этой опции использовать уравнения осесимметричной теории пьезоэлектричества, значительно более сложные, чем уравнения плоской деформации.

Построение твердотельной модели меридионального сечения пьезоэлектрического диска

Чаще всего твердотельная модель исходной области со сложной геометрией строится «снизу вверх», начиная с построения наиболее простых сущностей (Entities) – опорных точек (Keypoints) и заканчивая построением областей (Areas) для двумерных задач или объемов (Volumes) для трехмерных задач. Для построения опорных точек используется команда **K**, первым аргументом которой является номер точки, а остальными – координаты точки. Например, команда **K, 2, R1, -H/2** создает точку с номером 2 и координатами $x=R1$, $y=-H/2$, $z=0$ (значение по умолчанию). Для построения линии (сущность Line) между двумя точками используется команда **L**. Например, команда **L, 2, 3** создает линию в текущей системе координат между точками с номерами 2 и 3. При этом создаваемые линии нумеруются в программе автоматически, начиная с наименьшего доступного номера. Для построения областей можно использовать команды **A** и **AL**, при этом создаваемые области также нумеруются автоматически. Команда **AL** строит область по указанным линиям, перечисленным в порядке обхода по или против часовой стрелки (можно перечислить не более 10 линий), а команда **A** аналогично строит область по указанным опорным точкам (можно перечислить не более 18 точек). При этом, если между двумя точками определена линия, то команда **A** будет использовать ее при построении области, в противном случае в текущей координатной системе будет построена недостающая линия.

Согласно рис. 2, в меридиональном сечении диска две зоны с различным направлением поляризации, поэтому твердотельную осесимметричную модель диска можно построить состоящей из двух областей. На рис. 3 показаны построенные области A1 и A2 с нанесенными номерами областей и опорных точек. (Пункты меню Plot->Areas, для нумерации областей и точек PlotCtrls->Numbering->отметить Area numbers, Keypoint numbers)




Рис. 3 Нумерация областей модели меридионального сечения диска




Рис. 4 Линии с заданным разбиением и опорные точки модели меридионального сечения диска

Линии и точки, составляющие области A1 и A2, можно увидеть на рис. 4 (пункты меню Plot->Lines (здесь предварительно в Select->Entities выбраны линии L1-L9), для показа нумерации линий и точек PlotCtrls->Numbering->отметить Line numbers, Keypoint numbers). Здесь прямоугольная область A2 построена обычным образом из четырех линий L3, L4, L5, L9 и четырех точек 3, 4, 5, 6, поскольку верхняя и нижняя границы данной область совпадают с линиями расположения электродов. Область A1 состоит из шести точек 1, 2, 3, 6, 7 и шести линий L1, L2, L9, L7, L7, L8. Для этой области как верхнюю, так и нижнюю границы нужно строить из двух линий, причем опорные точки линий L1 и L7, моделирующих электроды, должны совпадать с концами электродов. Это требуется для того, чтобы при построении конечно-элементной сетки на концы электродов попали узлы конечных элементов. Следует обратить внимание на то, что линия L9 является общей для смежных областей A1 и A2.

Задание направления вектора поляризации для областей пьезоэлектрического диска.

Для плоских и осесимметричных двумерных задач удобно считать ось Oy осью предварительной поляризации в рабочей плоскости Oxy , поэтому в ANSYS для пьезоматериалов с различными векторами поляризации достаточно задавать для соответствующих областей элементные системы координат, повернутые нужным образом относительно глобальной системы координат.




Рис. 5 Направление векторов поляризации для областей меридионального сечения диска

В рассматриваемом примере направление вектора поляризации в области A1 совпадает с направлением оси Oy глобальной декартовой системы координат, поэтому для конечных элементов этой области не требуется вводить специальную элементную систему координат. Для области A2 направление вектора поляризации не совпадает с направлением оси Oy глобальной декартовой системы координат, поэтому для этой области требуется задать элементную систему координат с осью Oy , направленной вдоль вектора поляризации. Для этого элементную систему координат нужно повернуть на 180 градусов относительно глобальной декартовой системы координат. Следующий блок команд создает локальную систему координат и задает ее в качестве элементной для области A2.

```
LOCAL,11,0,,,180
```

```
ASEL,S,AREA,,2
```

```
AATT,1,,1,11
```

На рис. 6 зеленым цветом показаны направления оси Oy элементных систем координат. (Пункты меню Plot->Elements, для показа зон поляризации разным цветом PlotCtrls->Numbering->Elem/Attrib numbering-> выбрать Element CS num, Numbering shown

with ->Colors only, для показа направления элементной системы координат на каждом элементе PlotCtrls->Symbols, отметить ESYS Element coordinate sys)




Рис. 6 Элементные системы координат в областях меридионального сечения диска

Построение конечно-элементной сетки с регулярным разбиением.

Поскольку элемент **PLANE13** является четырехугольным конечным элементом и моделируемая область меридионального сечения диска является четырехугольником, а конкретнее, прямоугольником, то логично использовать *регулярное* конечно-элементное разбиение на четырехугольники (даже – на прямоугольники).

В ANSYS существует два типа конечно-элементного разбиения: регулярное (Mapped Mesh) и свободное (Free Mesh). *Свободным* называется разбиение, которое не имеет ограничений на форму элементов и не имеет заданного шаблона сетки. Обычно свободное разбиение используется для областей и объемов сложной формы. Для свободного разбиения достаточно задать средний размер конечного элемента. В противоположность свободному разбиению *регулярное разбиение* определяется заданной формой конечных элементов (например, только треугольники или только четырехугольники в случае разбиения областей) и заданным шаблоном конечно-элементной сетки с четкими «рядами» элементов. Регулярное разбиение выгодно использовать для областей и объемов простой формы, использование регулярного разбиения помогает значительно сократить время расчетов.

Для того чтобы построить регулярное разбиение четырехугольной области на четырехугольные конечных элементы, область должна удовлетворять двум критериям: 1) состоять из четырех линий, 2) противоположные стороны области должны быть разбиты на одинаковое число частей. Для области, состоящей из более, чем четырех линий, следует использовать конкатенацию смежных линий, чтобы сократить общее число линий до четырех.

В рассматриваемом примере меридионального сечения диска в принципе возможно два способа построения областей твердотельной модели. Например, можно было построить дополнительную линию можно точками 2 и 7 и составить три прямоугольных области, каждая из которых состояла бы из четырех линий.

Принятый здесь способ позволяет обойтись двумя областями, при этом область A2 является регулярной, а область A2 – нерегулярной, так как состоит из шести линий. Для превращения области A1 в регулярную достаточно объединить между собой линии L1 и L2, которые составляют нижнюю границу области, а также линии L6 и L7, составляющие верхнюю границу области. Для этого служит команда **LCCAT**, которая объединяет две линии в одну для последующего обеспечения регулярного разбиения. Если требуется объединить более двух линий, то в качестве аргумента команды **LCCAT** следует задать ALL, при этом будут объединены все линии выбранные ранее с помощью команды выбора линий **LSEL**. Следует помнить, что для построения регулярной конечно-элементной сетки суммарное число делений противоположных линий области должно быть одним и тем же! На рис. 5 показан результат объединения линий верхней и нижней

границ области A1. Полученные в результате линии L10 и L11 сохраняют заданное разбиение составляющих их линий (см. рис. 6).




Рис. 7 Конкатенация линий модели меридионального сечения диска

Задание граничных условий.

1) **Граничные условия на электродированных поверхностях.** В теории пьезоэлектричества пьезоэлектрические материалы рассматриваются как активные поляризующиеся диэлектрики. Части их границ Γ_{Vm} ; $m=1,2,\dots,M$; могут быть металлизированными поверхностями или электродами. Электроды Γ_{Vm} считаются эквипотенциальными поверхностями, т.е. на них электрический потенциал $\varphi(\mathbf{x},t)$ не должен зависеть от \mathbf{x} :

$$\varphi = V_m(t); \quad \mathbf{x} \in \Gamma_{Vm} \quad (2.1)$$

Граничные условия (2.1) являются главными или кинематическими электрическими граничными условиями для краевых задач пьезоэлектричества, если $V_m(t)$ – известные функции. Электроды с известными $V_m(t)$ называют электродами, запитываемые генераторами напряжений, а при $V_m = 0$ – короткозамкнутыми заземленными электродами.

Другой тип электродов представляют электроды Γ_{Ql} , запитываемые генераторами токов I_l . Для электродов Γ_{Ql} ; $l=1,2,\dots,L$; граничные условия имеют вид:

$$\varphi = \Phi_l(t); \quad \mathbf{x} \in \Gamma_{Ql} \quad (2.2)$$

$$\int_{\Gamma_{Ql}} \mathbf{n} \cdot [\mathbf{D}] d\Gamma = Q_l; \quad \dot{Q}_l = \pm I_l$$

где \mathbf{n} – нормаль к Γ_{Ql} ; $[\mathbf{D}]$ – скачок вектора электрической индукции \mathbf{D} ; Q_l – суммарный поверхностный заряд на электроде Γ_{Ql} ; причем значения потенциалов $\Phi_l(t)$ изначально не известны. При $Q_l = 0$ электроды (2.2) называются разомкнутыми или пассивными электродами.

Граничные условия (2.1) и (2.1) можно реализовать в ANSYS следующим образом. Узлы КЭ сетки, принадлежащие отдельному электроду Γ_{Vm} или Γ_{Ql} , свяжем в один узел (coupled DOF) командой **CP**. Далее, для выбранного по **CP** представителя – узла с номером N_VOLT можно либо задать значения потенциала $VM=V_m$ по команде **D,N_VOLT,VOLT,VM**, либо определить суммарный заряд $QL=Q_l$ командой **F,N_VOLT,AMPS,QL** (именно **AMPS**, а не **CHRG**). Блок команд для определения группы узлов и задания значения потенциала на первом электроде выглядит следующим образом:

```
NSEL,S,LOC,Y,-H/2
NSEL,R,LOC,X,0,R1
CP,1,VOLT,ALL
*GET,N_VOLT1,NODE,,NUM,MIN
D,N_VOLT1,VOLT,-VINP
```

Если требуется задать свободный электрод, то достаточно определить группу командой **CP**, но не использовать команду **D**.

2) «Механические» граничные условия. В рассматриваемом примере нужно поставить условие симметрии относительно оси вращения (оси Oy) и условие закрепления средней крайней точки по оси Oy .

Ограничение на степени свободы (для элемента PLANE13 их три: UX, UY, VOLT) можно задать либо с помощью команды **D**, которая задает ограничение на степени свободы на узлах, либо с помощью команды **DL**, которая задает ограничение на степени свободы на линиях. Следует иметь в виду, что все «твердотельные» граничные условия будут преобразованы в «конечно-элементные» граничные условия на этапе решения задачи. При этом, «твердотельные» граничные условия на заданной линии имеют приоритет над «конечно-элементными» граничными условиями на узлах той же линии.

2.1) **Условие симметрии** можно задать как с помощью команды **DL**, так и с помощью команды **D**. В команде **DL** можно использовать опцию SYMM (как это сделано в примере: **DL,8,,SYMM**). Другой способ состоит в том, чтобы выбрать нужные узлы и воспользоваться командой **D**. Выбрать узлы, лежащие на линии L8, можно с помощью команд:

```
LSEL,S,LINE,,8
```

```
NSLL,S,1 ! Выбор узлов, принадлежащих выбранным линиям
```

! 2й аргумент – ключ выбора внутренних узлов (здесь 1 – выбор внутренних и внешних узлов)

Команда **D,ALL,UX,0** задает для всех выбранных узлов равенство нулю компоненты вектора перемещений UX, что соответствует симметрии относительно оси Oy . Аналогично команда **D,ALL,UY,0** будет задавать для всех выбранных узлов равенство нулю компоненты вектора перемещений UY, что будет соответствовать симметрии относительно оси Ox .

2.2) **Условие жесткого закрепления** предполагает, что для заданных узлов вектор перемещений равен нулю, т.е. в случае плоской задачи $UX=0$ и $UY=0$. Соответствующий блок команд будет иметь вид

```
D,ALL,UX,0
```

```
D,ALL,UY,0
```

Конечно-элементная модель пьезоэлектрического диска с граничными условиями показана на рис. 8. (Пункты меню Plot->Elements, для отображения граничных условий PltCtrls->Symbols->отметить All applied BC)




Рис. 8 Конечно-элементная сетка с граничными условиями

Просмотр результатов.

В меню ANSYS это можно сделать, используя, например, следующие перемещения по меню:

General Postproc → Plot Results → Deformed Shape (деформированная форма)

General Postproc → Plot Results → Contour Plot → Nodal Solu... → DOF Solution →

X-Component of displacement (для изображения картины распределения перемещений U_x)

Y-Component of displacement (аналогично для изображения картины распределения перемещений U_y)

Displacement vector sum (для вывода модуля вектора перемещений)

Electric potential (для вывода распределения электрического потенциала)

General Postproc → Plot Results → Vector Plot → Predefined

DOF solution → Translation U (для вывода распределения вектора перемещений)

Flux & gradient → Elec field EF (для вывода распределения вектора электрического поля)

Сохранить полученный рисунок можно несколькими способами:

- PlotCtrls->Capture Image. Рисунок будет открыт в новом окне. Пункты меню File->Save as... позволят сохранить рисунок в выбранной папке без инвертации фона как графический файл с расширением .bmp).
- Plot Ctrls->Hard Copy->To File. Будет открыто окно Graphics Hard Copy. Здесь можно указать цветовую шкалу (Monochrome, Gray Scale, Color) расширение графического файла (.bmp, postscript, .tiff, .jpeg, .png), отметить Reverse Video для инвертации черного и белого цветов, задать имя файла. При этом рисунок будет сохранен с инвертацией фона с черного на белый в рабочей директории ANSYS, указанной при запуске программы.
- PlotCtrls->Write Metafile. Рекомендуется выбрать Invert White/Black для инвертации фона с черного на белый. Рисунок можно сохранить в выбранной папке как метафайл файл с расширением .emf или .wmf.

Результаты расчетов показаны на рис. 9-14.

DISPLACEMENT
 STEP=1
 SUB =1
 TIME=1
 DMX =.793E-09




Рис. 9 Картина деформированной формы пьезоэлектрического диска

MODEL SOLUTION
 STEP=1
 SUB =1
 TIME=1
 UX (AVG)
 ESYS=6
 DMX =.793E-09
 SMN =-.751E-09
 SMX =.923E-09




Рис. 10 Распределение перемещений UX

MODEL SOLUTION
 STEP=1
 SUB =1
 TIME=1
 UY (AVG)
 ESYS=6
 DMX =.793E-09
 SMN =-.263E-09
 SMX =.263E-09




Рис. 11 Распределение перемещений UY

```

VECTOR
STEP=1
SUB =1
TIME=1
O
MODE=442
MIN=0
MAX=.793E-09
    
```




Рис. 12 Распределение вектора перемещений

```

MODAL SOLUTION
STEP=1
SUB =1
TIME=1
VOLT [AVG]
KSYS=0
DMX =.793E-09
SMN =-1
SMX =1
    
```




Рис. 13 Распределение электрического потенциала

```

VECTOR
STEP=1
SUB =1
TIME=1
EF
ELEM=17
MIN=37.9599
MAX=1932.38

```




Рис. 14 Распределение вектора электрического поля

Индивидуальные задания

Написать программу на языке APDL ANSYS для расчета статического деформирования пьезоэлектрического преобразователя с многоэлектродным покрытием в форме диска (осесимметричная задача) или длинного цилиндра с поперечным сечением заданной формы (задача плоской деформации). Построить регулярную конечно-элементную сетку. Вывести результаты расчетов (деформированную форму, распределение перемещений, электрического потенциала, вектора электрического поля). Проанализировать результаты и оформить отчет.

Требования к отчету.

Отчет должен содержать ФИО студента, полное описание задачи, а также результаты, полученные с помощью конечно-элементного комплекса ANSYS:

- Регулярная конечно-элементная сетка с граничными условиями и показом элементной системы координат для каждого элемента (проверка направления поляризации)
- Деформированная форма пьезопреобразователя
- Картина распределения перемещений по оси OX
- Картина распределения перемещений по оси OY
- Картина распределения модуля вектора перемещений
- Картина распределения электрического потенциала
- Картина распределения вектора электрического поля

Варианты заданий

Значение потенциала на электродах: положить $V = 10$ В для нечетных номеров вариантов, $V = 5$ В для четных номеров вариантов.

№	Схема	Геометрические размеры
1		<p>Осесимметричная задача $R = 1 \text{ см}$ $h = 0.5 \text{ мм}$</p>
2		<p>Осесимметричная задача $R = 1 \text{ см}$ $h = 1 \text{ мм}$</p>
3		<p>Осесимметричная задача $R = 1.5 \text{ см}$ $h = 1 \text{ мм}$</p>
4		<p>Плоская деформация $a_1 = 1 \text{ см}$ $a_2 = 1.5 \text{ см}$ $h_1 = 1 \text{ мм}$ $h_2 = 0.5 \text{ мм}$</p>
5		<p>Плоская деформация $A = 1.5 \text{ см}$ $a = 0.5 \text{ см}$ $h_1 = 1 \text{ мм}$ $h_2 = 0.5 \text{ мм}$</p>

6		<p>Осесимметричная задача $R = 1.2 \text{ см}$ $h = 1 \text{ мм}$</p>
7		<p>Плоская деформация (половина модели) $a_1 = 1 \text{ см}$ $a_2 = 1 \text{ см}$ $h = 2 \text{ мм}$</p>
8		<p>Осесимметричная задача $a_1 = 0.3 \text{ см}$ $a_2 = 0.2 \text{ см}$ $h_1 = 0.5 \text{ мм}$ $h_2 = 0.5 \text{ мм}$</p>
9		<p>Плоская деформация $a_1 = 0.5 \text{ см}$ $a_2 = 0.6 \text{ см}$ $h_1 = 0.5 \text{ мм}$ $h_2 = 1 \text{ мм}$</p>
10		<p>Осесимметричная задача $a_1 = 0.4 \text{ см}$ $a_2 = 0.3 \text{ см}$ $h_1 = 1 \text{ мм}$ $h_2 = 1.5 \text{ мм}$</p>